
      program  dsp10020
c
c
c     2000/11/21 (Tue)   S.Kizu
c
c
      parameter  (ndmax=365*24, nd=ndmax, ndpp=480)
      integer     pgopen, idt(ndmax)
      real        xd(ndpp), yd(ndpp)
      character   fnam*4, pnam*45
c
c
c     --- 観測地点番号を入力する
      write(*,*)  ' station number  ?  (ex. HD13)  '
      read (*,'(a)')   fnam
c
c
c     --- データファイルを読み込む
      open ( 10, file='./data/'//fnam//'.hr', status='old' )
      read ( 10, '(a45)' )  pnam
c      write(*,*)  '  SAMPLING FREQUENCY (PER DAY) = ',ifq
      write(*,*)  '  NUMBER OF DATA               = ',nd
      nday=nd/24
      do j=1,nday
         is=(j-1)*24+1
         ie= j*24
         read ( 10, '(i6,24i5)', end=90 )   idate,(idt(i),i=is,ie)
      enddo
 90   close ( 10 )
c
c
c     --- 縦軸のレンジ
      ymin=  0.
      ymax=500.
c
c     --- 表示するページ数の計算
      npage=nd/ndpp
      if (npage*ndpp.lt.nd)  npage=npage+1
c
      if (pgopen('?').lt.1)  stop
      call pgscf ( 2 )
      call pgsch ( 1.2 )
c     --- ４８０個(＝２０日分)のデータを１ページとしてグラフ表示する
      do ipage=1,npage
         call pgpage
         is=(ipage-1)*ndpp+1
         ie= min(ipage*ndpp,nd)
         np=ie-is+1
         xmin=(ipage-1)*ndpp+1
         xmax= ipage*ndpp
c        --- ビューポートとウィンドウを設定する
         call pgsvp  ( 0.15, 0.9, 0.2, 0.85 )
         call pgswin ( xmin, xmax, ymin, ymax )
c        --- 両軸とメモリを描く
         call pgbox ( 'BCGINTS1', 120., 5, 'BCGINVTS1', 50., 5 )
c        --- Ｘ軸のタイトル
         call pgmtxt ( 'B', 4.,  0.5, 0.5, '\frData Number' )
c        --- Ｙ軸のタイトル
         call pgmtxt ( 'L', 4.5, 0.5, 0.5, '\frSea Level (cm)' )
c        --- 図のタイトル
         call pgmtxt ( 'T', 2., 0.5, 0.5, '\fr'//pnam )
         do i=is,ie
            ii=i-is+1
            xd(ii)=i
            yd(ii)=idt(i)

program 文は、(メイン)プログラムに名前を付ける時に用いる。通常は無くても構わないし、また名前はファイル名と違っていても構わない。

Fortran のプログラムでは、各行の最初の６カラム（文字）は特別な意味を持つので、
コメント行（１カラム目が c になっている行）や継続行（後述）以外の普通の命令文は、その後の７カラム目から書き始めなければならない。

また、多くの Fortran のシステムでは、各行は ７２カラム目までしか認識されない。
それ以上に命令文が長くなってしまう場合は、次の行の６カラム目に空白以外の記号
（例えば & など）を書いた上で、前の文の続きを書くことができる。　
このような行を 『継続行』 と呼ぶ。

"c" で始まる文は『コメント行』と呼ばれ、実行されない。
メモや説明文を書き込むときや、一時的にプログラムの
一部を実行させたくない時などに使うと便利。

Fortran のプログラムで用いられる変数には、整数型、実数型、文字型ほか幾つかの『型』がある。すべての変数は、それらの型の何れかに属さなければならない。

integer 文、real 文、character 文は、それぞれ、整数型、(単精度)実数型、文字型の変数を定義するのに用いられる命令文で、『型宣言文』と総称される。定義された変数のうち、カッコ（）付きのものは配列を意味し、（）内の数字がその配列のサイズ（大きさ＝要素数）を表す。

配列でない変数は暗黙の型宣言（別項参照）を冒さない限り定義しなくても構わないが、配列変数は必ずプログラムの冒頭で定義しなければならない。

配列変数のサイズは、基本的には数字で定義される。しかし、この例のように、予め parameter 文で数値を変数に代入しておき、その変数で配列変数のサイズを定義することも可能である。サイズが同じ多数の配列変数を定義したい場合などは、この方法を使うとサイズを一遍に書き変えることが出来るので便利。

character 文の変数についているアスタリスク（*）とその後の数字は、それぞれの文字型変数の長さを表す（例えば fnam*4 は４文字文の文字列変数として fnam を定義することを意味する）。

write 文は、後に続く変数（ここでは '   ' 内の文字列）を出力する（書く）命令文で、
read 文は、逆に後に続く変数（ここでは fnam の値）を入力する（読む）ための命令文。

それぞれ、write、read の直後に続く (   ) 内の部分のうち、カンマ（，） で区切られた前の部分が 『どこへ（どこから）』 出(入)力するのかを表し、カンマの後の部分が 『どのような形式で』 出（入）力するのか（書式）を表す。

この例では、どちらも最初の 『どこから(どこへ)』 の部分がアスタリスク（*）になっているが、このアスタリスクは write 文の場合はモニタ画面を、read 文の場合はキーボードを、それぞれ常に意味する。 つまり、この場合は画面に ' station number  ?  (ex.HD13) ' という文字列を表示させ、その後でキーボードからある文字型変数を読み込ませることになる。

一方、後半の書式（入出力の形式を指定する部分）については複雑なので詳しい説明を割愛するが、アスタリスクは「適当な形式にお任せする」ことを意味し、それ以外は陽に希望の書式を指定したい場合に用いられる。ケースバイケースでどちらも用いられる。

// （スラッシュ２つ）は、文字列を連結する時に用いられる。　この部分では、前の部分でキーボードから入力させた文字列変数を使って open 文で開くべきデータファイルの名前を完成させるのに使っている。


read 文や write 文でデータを読み書き（入出力）できる相手は画面やキーボードだけではなく、ファイルに対しても同様にこれを行うことが出来る。　open 文は、このようにファイルに対して入出力を行う際に、それに先立ってファイルを定義し、開くための命令文である。

カッコ（　）内は、開きたいファイルの名前（file=...... の部分）や、そのファイルの形式などを指定する部分で、必要に応じて（この例では使われていない）様々なオプションがある。

最初の数字（この例では 10）は、file=........ 以下で定義したファイルを割り付ける番号で、一旦このように番号に割り付けてしまうと、close 文でその割付を解いてしまうまでの間は、その番号がそのファイルを一意に指すことになる。 したがって、後に続く read 文でも、このファイルからそれぞれデータが読み込まれる。　

番号として用いる数字は 10 でなくても構わない。　使うシステムにもよるが、だいたい３桁までの数字なら使えると思われる。　ただし、１桁の数字の中には予め特定の装置に割り付けられているものもあるので、普通は ２桁 か ３桁 にしておく方が良いかも知れない。

また、ある番号を open 文で特定のファイルに割り付けたら、それを close 文で解放するまでは、他のファイルを同じ番号に割り付けることは出来ない。　

同時に開くことの出来るファイルの数はシステムによって異なるが、２０や３０程度ならまず問題ないはずである。




do ....... enddo は、それらで囲った命令文を指定回数だけ
繰り返すために用いられる。この例では、字下げした３行の
命令が、nday 回繰り返される。

do ...... enddo の他、do ....... continue も同じ目的で用いられる。　
ただし、後者の場合は文番号が必要である。





木津 昭一



この end=.... は、read 文のオプションの一つで、END 指定子と呼ばれる。
ここでは

「10 番に割り付けてあるファイルからデータを読み込む過程で、もしファイルの終わりまで到達してしまったら文番号 90 番の命令文へ飛べ」

ということを意味する。　通常、ファイルの終わりを越えて読み込み動作を行おうとするとエラーになってしまうので、ここではそれを避けるためにこのような指定を付けてある。

このような変数の書き方を　『DO 形並び』　と呼ぶ。　この部分では、

　　　　　　　　idt(is), idt(is+1), ........., idt(ie-1), idt(ie) 

のように配列変数 idt の is 番目から ie 番目までの要素を、
idate に引き続いて読み込ませている。


木津 昭一
最も単純な代入文の形式である。　それぞれ、右辺の数値が左辺の変数に代入される。　右辺は変数でも構わないし、また少し前の部分にある  nday=nd/24  のように計算式でも構わない。


if 文は、条件判定をし、その結果に基づいて実行処理を分岐する場合に用いられる。　いくつかの形式があるが、この部分はそのうちの　『論理 if 文』　と呼ばれる書き方になっている。

カッコ（　　）内が条件式で、この場合は npage と ndpp の積が nd よりも小さいか（.lt. は less than の略）どうかを判定している。　この条件式の結果が真の場合 （つまり、 npage*ndpp < ndpp である場合）、後ろの命令が実行される。　判定の結果が偽であれば、何も実行されない。　


１～５カラム目までの間に書かれる数字は　『文番号』　で、プログラム内の別の場所からその命令文を参照する際に使われる。

この部分では、直前の read 文でファイルの終わりが検出された場合の飛び先を示している。

これ以降の部分は、PGPLOT による作図のための命令群である。　

各々の call 文で必要なサブルーチンを呼び出し、それぞれ作図に
必要な操作を行わせている。　ここでは簡単にしか説明しないが、
詳しくは PGPLOT のマニュアルを参照して頂きたい。

pgopen は、PGPLOT の開始を宣言するサブルーチン。
カッコ（　）内を ’？’　とすると、実行時に図の出力先（画面とか、印刷用ファイルとか）を指定できる。


pgpage は、改ページをさせるサブルーチン。　このサンプルプログラムでは、長い時系列を幾つものページに分けて表示させるようになっているので、各ページの作図を行う前に改ページ（厳密には新しいページの開始宣言）を行っている。


Fortran には様々な組み込み関数が用意されている。　この min は大小判定を行うための関数の一つで、ここではカッコ（　　）内の変数を比較して小さい方の数値を左辺の変数 ie に代入するのに用いている。


グラフを書く際には、初めに、（紙や画面の）どの場所に、どのような範囲のグラフを書くか指定しなければならない。　これを、それぞれ、『ビューポート』、『ウィンドウ』 と呼ぶ。

pgsvp は、このうちのビューポートを指定するためのサブルーチンで、左下隅を (0.0, 0.0)、右上を (1.0, 1.0) とするような２次元の座標で範囲を指定する仕組みになっている。
この例では、ｘ 方向が　0.15～0.9、 ｙ 方向が 0.2～0.85 であるような範囲が指定されている。 0.9-0.15 ＞　0.85-0.2 なので、幾分横長のグラフが描かれる。

一方、pgswn は、ウィンドウを指定するためのサブルーチンで、この例では、ｘ 方向については xmin～xmax （ページの進行に従い変わっていく）の範囲を、ｙ 方向については ymin(=0)～ymax(=500) の範囲をグラフに描かせることを意味する。


pgbox は、軸や目盛りを記入するために用いられるサブルーチンで、カッコ（　　）内の６つのパラメータのうち、最初の３つがｘ方向、後の３つがｙ方向の軸についての指定を表す。　

これらの指定の仕方で、軸を線形軸にしたり対数軸にしたり、あるいは軸の目盛りを外側に出したり内側に納めたり、あるいは目盛りにつける数値の形式を定めたりといったことが様々に指定できる。　詳しくは、PGPLOT のサブルーチンのリファレンスマニュアルを参照のこと。


pgmtxt は、ここでは縦横両軸や図のタイトルを付けるのに用いられている。　カッコ（　　）内の最初のパラメータが何処にタイトルを付けるかを指定していて、それぞれ

・ 'B' は bottom、すなわち図の下側を、
・ 'L' は left、すなわち図の左側を、
・ 'T' は top、すなわち図の上側を

表している。　また、文字列の頭に付いている \fr は書体を指定していて、この場合はローマン（Roman）字体でそれぞれの文字列が描かれる。

10 番のファイルから、pnam という 45 文字分の文字型変数を
読み込むことを意味する書式指定。　冒頭の変数の型宣言と
矛盾していると、エラーが起こる。


書式指定の i  は整数を意味する。　

ここでは、行の先頭から 6 桁を idate という整数型変数に読み込み、
それに続いて 5 桁の整数 24 個を idt という整数型の配列変数に
読み込ませている。

四角で囲った部分のメモ記号をダブルクリックすると、
その部分についての説明窓が現れます。　

説明窓を閉じる時は、窓の中でシングルクリックし、
さらにそれによって左上隅に現れる灰色の部分を
シングルクリックして下さい。

         enddo
c        --- 点を打つ
         call pgpt   ( np, xd, yd, 20 )
c        --- 線で結ぶ
         call pgline ( np, xd, yd )
      enddo
c
      call pgclos
c
      stop
      end

pgpt は、指定した場所に 『点を打つ』 （より正確には、『記号を描く』） 時に
使うサブルーチンである。　最初の引数( np )が記号の数を、２番目と３番目の
変数が各々その x，y 座標を、そして最後の引数（ 20 ）が記号の種類を表す。


pgline は、指定した座標を順番に線で結ぶときに用いられるサブルーチンである。

この例の場合は、

　　　(xd(1), yd(1))，　(xd(2), yd(2))，　．．．．．．．，　(xd(np), yd(np))

の np 個の点を順番に線で結ぶことを意味する。

線の種類や太さは、前もって指定することができるが、
無指定の場合は、既定値の 『細い実線』 が使われる。


pgclos は、PGPLOT による作図を終えることを宣言するサブルーチンである。
作図の最後に必ず call すること。

プログラムの最後には、必ず end 文を書くこと。 

stop 文は、プログラムの実行をそこで打ち切る命令で、プログラムの末尾のほか、
必要に応じて（たとえば、何らかの条件判定の結果によって処理を終了させるために）
途中に挿入しても構わない。

また、end 文の直前に限って、stop 文を省略しても構わない（この例の場合も）。

作図の後半部分。　

前の部分に引き続いて、データの位置に記号を描き、それらを線で結ぶ。　そして最後に pgｃlos を call して、作図を終了する。

